* Model individual turnout

version 13.1
clear all

* get the data

use querschnitt-2013.dta , replace

keep if ! nachwahl

do survey-features.do

* fully interactive model

qui svy: logit wahlbeteil i.male##i.distostwest i.bildung##i.distostwest i.pi##i.distostwest c.demzufriedenheit##i.distostwest (c.alter#c.alter c.alter)##i.distostwest i.religpractice##i.distostwest i.gewerkschaft##i.distostwest i.einpersonenhaushalt##i.distostwest c.interesse##i.distostwest  i.distostwest 

est store fullinteract

qui svy: logit wahlbeteil i.male i.bildung i.pi c.demzufriedenheit (c.alter#c.alter c.alter) i.religpractice i.gewerkschaft i.einpersonenhaushalt  c.interesse i.distostwest 

est store nurostwest

qui svy: logit wahlbeteil i.male i.bildung i.pi c.demzufriedenheit (c.alter#c.alter c.alter) i.religpractice i.gewerkschaft i.einpersonenhaushalt  c.interesse    

est store pure


est restore pure
getaicw
estadd scalar r(AICW)

est restore nurostwest
getaicw
estadd scalar r(AICW)

est restore fullinteract
getaicw
estadd scalar r(AICW) 
 


* Output
esttab fullinteract nurostwest pure , b(%9.1f) label nobaselevel nomtitle se  nodepvars interaction(" X ") varwidth(31) rename (_cons Konstante) stats(N AICW)
esttab fullinteract nurostwest pure using wahlbeteiligungsmodelle.rtf  , b(%9.1f) label nobaselevel nomtitle se  nodepvars interaction(" X ") varwidth(31) rename (_cons Konstante) stats(N AICW) rtf replace 
